

---

# PICCOLINE

---

*Myresnak*

---

---



PICCOLINE

Myresnak



# Myresnak

---

**Nøgleord:** PICCOLINE, RC759, LOGO, MYRESNAK

---

**Resumé:** MYRESNAK er et grafisk orienteret programmerings-  
sprog, som vil kunne anvendes både af børn og  
voksne. Sproget bygger på det amerikanske sprog  
LOGO, og er udviklet af Viggo Sadolin, Danmarks  
Lærerhøjskole

---

**Udgave:** Marts 1985

Copyright © 1985 A/S Regnecentralen af 1979  
RC Computer A/S  
Udgivet af A/S Regnecentralen af 1979, København

Brugere af denne manual gøres opmærksom på, at specifikationer  
heri uden forudgående varsel kan ændres af RC. RC er ikke an-  
svartlig for typografiske fejl eller regnefejl, som kan forekomme i  
denne manual, og er ikke ansvarlig for skader forårsaget af benyt-  
telsen af dette dokument.

**MYRESNAK**  
**BRUGERVEJLEDNING**

<u>INDHOLDSFORTEGNELSE</u>	<u>SIDE</u>
0. Introduktion .....	3
1. Installation af MYRESNAK .....	4
2. Opstart af MYRESNAK .....	6
2.1 Opstart direkte fra CCP/M .....	6
2.2 Automatisk opstart af MYRESNAK .....	7
2.3 MYRESNAK er klar .....	8
3. Anvendelse af MYRESNAK .....	9
3.1 Direkte tegning med myreordre .....	9
3.2 Myreprogrammer .....	11
3.3 Myreprogrammer på disketten .....	13
4. MYRESNAK kommandoer .....	15
4.1 Myreordre .....	15
4.2 Systemordre .....	19
5. Udskrivning på printer og plotter .....	22
6. Anvendelse af farveskærm .....	23
7. Forkortelser .....	24
8. Funktionstaster .....	25
9. Eksempler på MYRESNAK-programmer .....	26
10. Referencer .....	30

## 0. Introduktion.

MYRESNAK er et enkelt programmeringssprog, som vil kunne anvendes både af børn og voksne.

Idegrundlaget for MYRESNAK er det amerikanske sprog LOGO, som blev udviklet i 70'erne af en gruppe forskere under ledelse af Seymour Pappert. Baggrunden for LOGO beskrives i Seymour Pappert's bog 'Den totale Skildpad-detur' (ref.2.)

MYRESNAK er udviklet af Viggo Sadolin, Danmarks Lærerhøjskole. Viggo Sadolin beskriver i sin bog 'MYRESNAK' (Teknisk Forlag 1984) hvorledes MYRESNAK f.eks kan anvendes i skolens matematikundervisning på 6.-10. klassetrin.

Bogen kan anbefales som supplement til denne manual, idet den mere detaljeret beskriver anvendelsen af MYRESNAK.

MYRESNAK er implementeret på skoledatamaten RC PICCOLINE af A/S Regnecentralen.

1. INSTALLATION AF MYRESNAK.

På 'MYRESNAK'-distributionsdisketten, som leveres fra A/S REGNECENTRALEN, vil der være følgende filer :

MYRESNAK.CMD      (MYRESNAK programmet )  
  
POLYGON.MYR      (Eksempler på MYRESNAK-programer)  
  SPIRAL.MYR  
  HCIRKEL.MYR  
  VCIRKEL.MYR  
  STRAALE.MYR  
  SOL.MYR  
  HJUL.MYR  
  POLY.MYR  
POLYSTEP.MYR  
  POLYTO.MYR  
  
STARTUP.0      (Anvendes ved installation af MYRESNAK)

Den originale MYRESNAK-Distributionsdiskette kopieres straks, og må derefter ikke benyttes. Fremgangsmåden for denne kopiering, kan læses i PICCOLINE BRUGERVEJLEDNING (ref.3).

Distributionsdisketten indeholder ikke CCP/M styresystemet, og opstart af MYRESNAK kan derfor ikke ske umiddelbart herfra.

Styresystemet CCP/M skal derfor overføres til MYRESNAK-disketten, før denne kan anvendes.

På denne skal ligeledes befinde sig det grafiske styresystem GSX, som skal være til stede på MYRESNAK arbejdsdisketten, før denne kan tages i brug.



Fremgangsmåden for installation af MYRESNAK kan f.eks. være følgende:

1. Start PICCOLINE på sædvanlig måde, d.v.s. ved hjælp af CCP/M-systemdisketten i diskettestation 1.
2. Indsæt kopien af MYRESNAK-disketten i diskettestation 2.
3. Tryk på ESC til hovedmenuen.  
På spørgsmålet : 'OK at vende tilbage til TMP (j/n)' svares j.
4. Skriv nu følgende :

```
PIP B:=A:CCPM.SYS
PIP B:=A:GRAPHICS.CMD
PIP B:=A:ASSIGN.SYS
PIP B:=A:DD*.SYS
```

Herved kopieres de nødvendige systemfiler over på MYRESNAK-disketten.

5. Såfremt det ikke ønskes at systemet automatisk skal starte MYRESNAK, skrives til slut følgende :

```
ERA B:STARTUP.0
```

Flytning af CCP/M-filerne kan også ske ved hjælp af programmet FILADM. Dette er beskrevet i PICCOLINE BRUGERVEJLEDNING/BETJENING (ref. 3), samt i PICCOLINE BRUGERVEJLEDNING/INSTALLATION OG VEDLIGEHOLDELSE (ref. 4).

I disse vejledninger kan man desuden læse hvorledes man kan lave automatisk opstart af MYRESNAK, ved hjælp af PICCOLINE MENUSYSTEMET.

2. OPSTART AF MYRESNAK.

Opstart af MYRESNAK kan ske direkte fra CCP/M styresystemet, eller automatisk ved opstart af PICCOLINE.

2.1 OPSTART FRA CCP/M.

Før MYRESNAK kan indlæses fra disketten skal det grafiske styresystem GSX være indlæst. Dette gøres således:

A>GRAPHICS

Herefter indlæses MYRESNAK ved at skrive følgende programnavn :

A>MYRESNAK

Disse to kommando'er kan fra version 2.2 (jan.85) af PICCOLINE distributionsdisketten, slås sammen på samme linie således :

A>GRAPHICS//MYRESNAK

Herved indlæses først det grafiske styresystem GSX, og derefter MYRESNAK.

Såfremt det grafiske styresystem ikke er indlæst når MYRESNAK startes, fremkommer denne fejlmelding på skærmen :

Myresnak, Version 1.1 Mar. 85. Løbenummer RCXXXXXXXX

FEJL: GSX-grafik delen er ikke installeret.  
Dette må gøres før Myresnak kan bruges.

## 2.2 AUTOMATISK OPSTART AF MYRESNAK.

Automatisk opstart af MYRESNAK kan ske, ved at udnytte det avancerede MENUSYSTEM, der ligger på PICCOLINE distributionsdisketten. Beskrivelse af dette kan findes i PICCOLINE BRUGERVEJLEDNING (ref.3).

En anden måde at få MYRESNAK til at starte automatisk, er at benytte filen STARTUP.0, som ligger på MYRESNAK-distributionsdisketten.

STARTUP.0 indeholder følgende kommandosekvens :


```
GRAPHICS//MYRESNAK//GRAPHICS NO
```

Ved systemopstart vil først det grafiske styresystem GSX blive indlæst, derefter MYRESNAK, og når MYRESNAK forlades, slettes GSX igen.

### 2.3 MYRESNAK ER KLAR.

Når MYRESNAK er startet, vil der på dataskærmen fremkomme en melding om, at MYRESNAK er klar til brug.

Teksten på dataskærmen vil forsvinde efter kort tid, eller såsnart tastaturet røres, og følgende skærbilleder vil fremkomme :



Denne side kaldes TEGNESIDEN, og det er her MYRESNAK udfører sine tegninger. Den lille trekant i midten er MYREN, og myrens spids viser tegneretningen.

På TEGNESIDEN vil 'myren' kunne bevæge sig, og bevægelsens længde angives i antal 'myreskridt'. TEGNESIDEN måler ca. 300 skridt vandret, og ca. 200 skridt lodret.

Nederst på tegnesiden, er der to linier hvor man kan skrive MYRESNAK-kommandoer.

På den nederste linie fremkommer en blinkende markør (en lille streg), og dette betyder at indtastning af MYRESNAK-kommandoer nu kan begynde.

### 3. ANVENDELSE AF MYRESNAK.

MYRESNAK kan anvendes på to måder, enten ved at afgive kommandoer, som udføres direkte, eller ved at skrive MYRESNAK programmer. Ved indtastning af MYRESNAK-kommandoer findes to typer, MYREORDRER og SYSTEMORDRER.

MYREORDRER anvendes til at styre 'myren', hvorimod SYSTEMORDRER anvendes af MYRESNAK til f.eks at gemme programmer på disketten, udskrive på printeren, indskrive og redigere programmer m.v.

#### 3.1 DIREKTE TEGNING MED MYREORDRER.

Når MYRESNAK startes befinder vi os på tegnesiden. Tegnesiden består af en tegneflade, samt to linier nederst på skærmen. På den nederste linie kan man skrive myreordrer som udføres direkte på skærmen. Myren er vist som en lille trekant midt på skærmen, og trekantens spids (myrens næse) peger i tegneretningen. Efter udførelse af en myrekommando, vil kommandoen blive flyttet op i den øverste af de to nederste linier. Herefter kan næste myreordre udføres.

Efter hver myreordre skal man huske at trykke på RETURN tasten (den grønne tast med knækpil).

Eksempel :           FREM(50)  
                      VDREJ(90)  
                      FREM(100)  
                      VDREJ(90)  
                      FREM(50)  
                      VDREJ(90)  
                      FREM(100)  
                      VDREJ(90)

Resultatet skulle gerne blive et rektangel.

Hvis man ønsker at slette billedet på skærmen, skrives myreordren RENS. Denne ordre anbringer myren i sin startposition, med næsen mod nord (op).

Som kommandoer til direkte udførelse, kan følgende anvendes :

BAK	BK	BLÅ	ERYM
FLYV	FR	FREM	GRØN
HD	HDREJ	HJEM	HVID
KRAVL	MYRE	NORD	RENS
RØD	SNOR	SORT	START
STRAM	VD	VDREJ	

Herudover kan alle SYSTEMORDRER anvendes.

En beskrivelse af de enkelte MYREORDRER og SYSTEMKOMMANDOER findes i kap.4.

Såfremt der anvendes 'ulovlige' ordrer eller kommandoer, vil disse blive ignoreret. MYRESNAK vil normalt ikke fremkomme med en fejlmelding, men blot give et BIP som signal om at noget er galt.

### 3.2 MYREPROGRAMMER.

En anden måde at anvende MYRESNAK, er at skrive MYRESNAK programmer. Disse kan gemmes på disketten til senere brug.

Før et program kan indskrives skrives kommando'en HUSK

Eksempel :                   HUSK

MYRESNAK forlader nu tegnesiden, og går over på tekst-siden, hvor man kan indskrive sit program.

Programmet indledes med et programnavn, som evt. kan have nogle parametre, dvs. værdier som skal bruges i programmet. Som parametre anvendes enkeltbogstaver i en parentes efter programnavnet. Er der flere parametre, skal disse adskilles af komma.

Eksempler på programnavne :

```
TREKANT
SPIRAL
KASSE(L,B)
DULLE(A,V,T)
```

Et program afsluttes med sætningen SLUT, og kan ialt maksimalt bestå af 20 linier.

Eksempel :                   FIRKANT  
                              GENTAG(4)  
                              FREM(50)  
                              VDREJ(90)  
                              HERTIL  
                              SLUT

Eksempel :           KASSE(L,B)  
                  FREM(L)  
                  VDREJ(90)  
                  FREM(B)  
                  VDREJ(90)  
                  FREM(L)  
                  VDREJ(90)  
                  FREM(B)  
                  VDREJ(90)  
                  SLUT

Mens programmet indskrives kan der rettes ved hjælp af de fire markørpile, samt tasterne 'slet tegn', 'tegn ind', A1 (indsæt linie) og A2 (slet linie).

Når programmet er færdigindtastet, trykkes på ESC. Herefter kan programmet udføres ved blot at skrive programmets navn.

Eksempel :           FIRKANT.

Programmet FIRKANT vil nu indgå i 'myrens ordforråd', på samme måde som de indbyggede myreordrer.

Eksempel :           MANGEKANT  
                  GENTAG(20)  
                  FIRKANT  
                  VDREJ(18)  
                  HERTIL  
                  SLUT

Hvis der i programmet indgår parametre, skal disse angives når programmet kaldes.

Eksempel :           KASSE(50,20)



Et MYRESNAK-program vil kunne kalde sig selv, dvs. benytte sit eget programnavn inde i programmet.

Eksempel :           POLY(S,V)  
                      FREM(S)  
                      HDREJ(V)  
                      POLY(S,V)  
                      SLUT

MYRENSNAK vil kunne rumme ialt 10 programmer samtidigt, blot disse har forskellige navne.

Kommandoen KATALOG vil give en oversigt over de MYRESNAK-programmer, som befinder sig i arbejdshukommelsen.

Ønskes et program udskrevet på dataskærmen, gøres dette ved hjælp af kommandoen LIST. Efter LIST-kommandoen spørger MYRESNAK om programmets navn, og her svares f.eks FIRKANT.

Skal programmet rettes benyttes på samme måde kommandoen RET.

Et program kan slettes fra arbejdshukommelsen, ved at bruge kommandoen GLEM. MYRESNAK vil spørge om navnet på det program, som skal 'glemmes'

### 3.3 MYRESNAK-PROGRAMMER PÅ DISKETTEN.

Når man indskrifer MYRESNAK-programmer, gemmes disse i arbejdshukommelsen. Slukkes der for PICCOLINE, vil programmerne blive slettet.

Ønskes programmerne gemt til senere brug, kan disse gemmes på disketten. Dette gøres ved at skrive kommandoen GEM, hvorefter MYRESNAK spørger om navnet på det program man ønsker at gemme. På disketten tillades kun programnavne på indtil 8 bogstaver. Desværre tillader CCP/M ikke at man anvender Æ eller Å i programnavne. For programmer navne længere en 8 tegn, gemmes kun de første 8 tegn.

I CCP/M's katalog over filer på disketten, gemmes programmer fra MYRESNAK, med betegnelsen MYR. Et MYRESNAK-program vil derfor under CCP/M f.eks se således ud:  
FIRKANT.MYR

I MYRESNAK kan man få en oversigt over MYRESNAK-programmer på disketten ved at skrive BIBLIOTEK, eller forkortet BB.

Når et MYRESNAK-program skal indlæses fra disketten, bruges systemordren HENT. MYRESNAK spørger herefter om navnet på det program som skal indlæses. Her angives kun programmets navn i MYRESNAK, og ikke det fulde CCP/M programnavn, dvs. .MYR skal ikke medtages.

Sletning af et MYRESNAK-program på disketten, sker ved anvendelse af kommandoen SLET. MYRESNAK spørger selv om navnet på det program, som skal slettes.

#### 4. MYRESNAK KOMMANDOER

I MYRESNAK indgår to typer kommandoer, nemlig MYREORDRER og SYSTEMORDRER.

MYREORDRER er ordrer som myren forstår, og som benyttes til at styre myren, dvs. tegne på dataskærmen eller på printer eller plotter.

SYSTEMORDRER benyttes af MYRESNAK, og har ikke direkte noget med myrens bevægelser at gøre.

##### 4.1 MYREORDRER.

BAK : Myren bakker det antal skridt som angives, f.eks BAK(20)

BK : Forkortelse for kommandoen BAK.

BLÅ : Tegnefarven ændres til blå (kun farveskærm).

ERYM : Efter angivelse af denne kommando, forsvinder myren. Dette bevirker at tegnehastigheden øges. Se iøvrigt MYRE.

FLYV : Myren letter fra overfladen, og tegner derefter ikke noget spor på skærmen. Myren kan dog fortsat bevæges rundt på tegnefladen.

FR : Forkortelse for kommandoen FREM.

- FREM : Myren bevæger sig det antal skridt som angives efter FREM. Eks. FREM(50).
- GENTAG : Anvendes når en del af programmet skal gentages.  
Efter ordet GENTAG angives hvor mange gange der skal gentages.
- Eksempel :       GENTAG(8)  
                  FREM(20)  
                  VDREJ(45)  
                  HERTIL
- GRØN : Ændrer tegnefarven til grøn.
- HD : Forkortelse for HDREJ.
- HDREJ : Myren drejer sin tegneretning det antal grader til højre, som er angiver efter HDREJ. F.eks HDREJ(60)
- HERTIL : Kommandoen anvendes som afslutning på en programdel som skal gentages. Se desuden GENTAG.
- HJEM : Myren sendes hjem til sin startposition, d.v.s i skærmens midte. Den tegning som står på skærmen slettes ikke.
- HVID : Sætter tegnefarven til hvid, dvs dataskærmens forgrundsfarve.

HVIS : Anvendes når man ønsker at kontrollere om en betingelse er opfyldt. Hvis denne betingelse er opfyldt, udføres sætningen som er angivet sidst i HVIS-linien. Hvis betingelsen ikke er opfyldt fortsætter programmet.

EKS. HVIS (A=0)FREM(50)

Som relationsoperator kan anvendes :

= : lig med  
<> : forskellig fra  
< : mindre end  
> : større end  
<= : mindre end eller lig med  
>= : større end eller lig med

KRAVL : Anvendes når man ønsker at myren skal af-sætte spor, dvs. tegne på skærmen. Hvis kommandoen FLYV har været anvendt, og man ønsker at myren igen skal tegne, anvendes KRAVL

MYRE : Myren (en lille trekant) vil efter denne kommando kunne ses på skærmen. Myrens spids peger i tegneretningen. Anvendes især under programudviklingen, idet tegneforløbet da vil kunne følges. Se iøvrigt ERYM

NORD : Myrens tegneretning vendes mod nord, dvs. op

RENS : Sletter tegnefladen (dataskærmen), og anbringer myren i midten med retning mod nord (dvs. op).

- RØD : Sætter tegnefarven til rød.
- SNOR : Myren fastgør en snor i det punkt hvor den befinder sig. Når kommandoen STRAM anvendes trækkes en streg fra det sted hvor snoren blev sat fast.
- SORT : Sætter tegnefarven til sort, dvs. data-skærmens baggrundsfarve.
- START : Anbringer myren i startstilling, uden at slette skærbilledet.
- STRAM : Strammer snoren, dvs. tegner en streg, fra det punkt hvor den blev sat fast, ved hjælp af kommandoen SNOR.
- VD : Forkortelse af kommandoen VDREJ.
- VDREJ : Myren drejer sin tegneretning det antal grader til venstre, som er angivet efter VDREJ. F.eks VDREJ(60)

4.3 SYSTEMORDRER.

- BB : Forkortelse for kommandoen BIBLIOTEK.
- BIBLIOTEK : Udskriver en liste over de MYRESNAK-programmer, som er på disketten.
- DISKETTE : Skal anvendes efter disketteskift.
- FARVER : Før farver kan anvendes, skal denne kommando bruges.
- GEM : Anvendes når et myreprogram skal gemmes på disketten. Inden programmet gemmes spørger MYRESNAK om et programnavn. Dette må højst være på 8 bogstaver.
- GLEM : Anvendes når et indtastet program skal slettes fra arbejdslageret. GLEM kommandoen sletter ikke programmer på disketten
- HENT : Anvendes til at hente et MYRESNAK-program på disketten. Myresnak spørger selv om programnavn.
- HJÆLP : Udskriver en liste over alle MYRESNAK kommandoer.

- HUSK : Anvendes når man ønsker at starte indskrivning af et MYRESNAK-program.
- KATALOG : Udskriver en liste over MYRESNAK programmer, som findes i arbejdsbureauet.
- KT : Forkortelse af kommandoen KATALOG
- LIST : Anvendes når man ønsker et MYREPROGRAM udskrevet på skærmen. Efter kommandoen LIST spørger MYRESNAK om programmets navn.
- PAPIR : Anvendes i forbindelse med udskrift på printer.
- PLOTTER : Anvendes i forbindelse med tegning på plotter.
- RET : Anvendes når man ønsker at rette i et MYRESNAK-program. MYRESNAK spørger selv om programmets navn.
- SKRIV : Udskriver det sidst anvendte program på printeren.
- SKÆRM : Anvendes når man ønsker tegning på dataskærmen. Ved opstart af MYRESNAK vælges altid dataskærmen som tegneflade.



- SLET : Når man ønsker at slette et program på disketten, anvendes kommandoen SLET. MYRESNAK spørger selv om programmets navn.
- SLUT : Kommandoen SLUT afslutter ethvert MYRESNAK program.
- SORTHVID : Kommandoen bruges hvis der skal tegnes på en monokrom (sorthvid) dataskærm.
- STOP : Stopper MYRESNAK, og returnerer til CCP/M.
- TEGNING : Skifter fra tekstsiden til tegnesiden.
- TG : Forkortelse af kommandoen TEGNING.
- TRYK : Anvendes ved tegning på printer.

5. TEGNING PÅ PRINTER OG PLOTTER.

I MYRESNAK kan tegninger tegnes på tre forskellige tegneenheder, nemlig dataskærm, plotter og printer. Når andet ikke specificeres vil tegningen altid fremkomme på dataskærmen.

Ønskes tegningen på printer, f.eks RC603P eller RC604P, sker dette ved først at give MYRESNAK-kommandoen :

PAPIR

Herefter angives de kommandoer eller programnavne som skal tegnes, og når dette er sket vil tegning på printeren starte ved at afgive kommandoen :

TRYK

Eksempel :       PAPIR  
                  FREM(50)  
                  VDREJ(90)  
                  FREM(100)  
                  VDREJ(90)  
                  FREM(50)  
                  VDREJ(90)  
                  FREM(100)  
                  VDREJ(90)  
                  TRYK

Eksempel :       PAPIR  
                  FIRKANT  
                  TRYK

Hvis dataskærmen igen skal benyttes, skrives systemordren SKÆRM, inden tegning påbegyndes.

Tegning på plotter foregår på samme måde, som tegning på printeren. Før tegningen påbegyndes skrives PLOTTER, og herefter skrives de myreordrer eller programnavne, som skal tegnes.

Eksempel :        PLOTTER  
                  FIRKANT

## 6. ANVENDELSE AF FARVESKÆRM.

Hvis datamaskinens dataskærm kan vise farver, vil denne tegne med den valgte forgrundsfarve, såfremt andet ikke ønskes. Valg af forgrundsfarve sker ved hjælp af et hjælpeprogram GKONFIG, som findes på CCP/M distributionsdisketten (ref.4)

Angives systemordren FARVER, vil det være muligt at vælge forskellige tegnefarver. De tilgængelige tegnefarver er BLÅ, GRØN, RØD, HVID (= dataskærmens forgrundsfarve) eller SORT (= dataskærmens baggrundsfarve).

Valg af tegnefarve sker ved at angive en af myreordrene BLÅ, GRØN, HVID, RØD eller SORT.

Skal der igen skiftes til en monokrom dataskærm ('sort/hvid'), sker dette ved myreordren SORTHVID.

Eksempel :        KASSER  
                  FARVER  
                  RØD  
                  FIRKANT  
                  VDREJ(90)  
                  BLÅ  
                  FIRKANT  
                  VDREJ(90)  
                  GRØN  
                  FIRKANT  
                  VDREJ(90)  
                  HVID  
                  FIRKANT  
                  SLUT

7. FORKORTELSER.

En række MYRESNAK-kommandoer, som anvendes ofte, kan forkortes. Dette gælder følgende :

<u>KOMMANDO</u>	<u>FORKORTEELSE</u>	<u>EKSEMPEL</u>
BAK	BK	BK(30)
FREM	FR	FR(40)
HDREJ	HD	HD(45)
VDREJ	VD	VD(90)
BIBLIOTEK	BB	
KATALOG	KT	
TEGNING	TG	

8. FUNKTIONSTASTER.

En række af de hyppigst anvendte MYRESNAK-kommandoer, er anbragt i PICCOLINE's funktionstaster.


Disse taster har, når MYRESNAK startes, følgende indhold:

F1	:	BIBLIOTEK
F2	:	HENT
F3	:	GEM
F4	:	KATALOG
F5	:	TEGNING
F6	:	LIST
F7	:	RET
F8	:	HUSK
F9	:	GLEM
F10	:	RENS
F11	:	MYRE
F12	:	ERYM
A1	:	<indsæt linie>
A2	:	<slet linie>
A3	:	<slet resten af linie>
A4	:	(undefineret)

9. EKSEMPLER PÅ MYRESNAK-PROGRAMMER.

Eksempel 1.


```
POLYGON(A,S)  
  GENTAG(A)  
  FREM(S)  
  VDREJ(360/A)  
  HERTIL  
SLUT
```



POLYGON(6,30)

Eksempel 2.

```
SPIRAL(S,V)  
  FREM(S)  
  HDREJ(V)  
  HVIS(S<100)SPIRAL(S+3,V)  
SLUT
```



SPIRAL(5,144)

Eksempel 3.

```
HCIRKEL(R)
  GENTAG(9)
    HDREJ(5)
      FREM(R*3.14159/18)
        HDREJ(5)
          HERTIL
        SLUT
```



HCIRKEL(30)

Eksempel 4.


```
VCIRKEL(R)
  GENTAG(9)
    VDREJ(5)
      FREM(R*3.14159/18)
        VDREJ(5)
          HERTIL
        SLUT
```



VCIRKEL(30)

Eksempel 5.


```
STRAALE(R)
  HCIRKEL(R)
  VCIRKEL(R)
  HCIRKEL(R)
  VCIRKEL(R)
  SLUT
```



STRAALE(20)

Eksempel 6.


```
SOL(R)
  GENTAG(9)
    STRAALE(R)
      HDREJ(160)
    HERTIL
  SLUT
```



SOL(20)

Eksempel 7.

```
HJUL(R)
  FLYV
  HDREJ(90)
  FREM(R)
  SNOR
  BAK(R)
  VDREJ(90)
  KRAVL
  GENTAG(36)
    HDREJ(5)
    FREM(R*3.14159/18)
    HDREJ(5)
  STRAM
  HERTIL
  SLUT
```




HJUL(40)



Eksempel 8.

```
POLY(S,V)
  FREM(S)
  HDREJ(V)
  POLY(S,V)
SLUT
```




POLY(100,156)

Eksempel 9.

```
POLYSTEP(S,V)
  FREM(S)
  HDREJ(V)
SLUT
```

```
POLYTO(S,V,T,U)
  POLYSTEP(S,V)
  POLYSTEP(T,U)
  POLYTO(S,V,T,U)
SLUT
```



POLYTO(50,90,80,135)

10. REFERENCER.

- (1) Viggo Sadolin : MYRESNAK  
Teknisk Forlag 1984.
- (2) Seymour Pappert : Den totale skildpaddetur  
(Mindstorms) GAD 1984.
- (3) A/S Regnecentralen : PICCOLINE brugervejledning
- (4) A/S Regnecentralen : PICCOLINE Installation og  
vedligeholdelse.